DECUS
PROGRAM LIBRARY

DECUS NO.	FOCAL8-88
TITLE	ATOMIC AND MOLECULAR TRANSITION PROBABILITIES IN FOCAL
AUTHOR	Harold Metcalf
COMPANY	Physics Department SUNY Stony Brook Stony Brook, New York
DATE	December 11, 1969
SOURCE LANGUAGE	FOCAL

ATOMIC AND MOLECULAR TRANSITION PROBABILITIES IN FOCAL

DECUS Program Library Write-up

DECUS No. FOCAL8-88

In our work on molecular spectroscopy, we find it necessary to calculate optical transition probabilities between select rotational states. We are interested in the matrix element

$$P = (\alpha' \; S' \; K' \; J' \; I' \; F' \; M' \;\; | \; T(k,q) \; | \; \alpha \; S' \; K \; J \; I' \; F \; M)$$

in the notation of Edmonds (Angular Momentum in Quantum Mechanics, Princeton University Press, Princeton, N.J. 1957). This can be written as the product of several calculable terms. We consider the case of dipole radiation so that T is the electric dipole operator, k = 1 since T is a vector, and q = ±1, 0. Then

$$P = \begin{pmatrix}\text{phase}\\ \text{factor}\end{pmatrix} \; (\text{constant}) \; (\alpha' \; K \;\; || \; T(k) \; || \; \alpha \; K) \begin{Bmatrix} K' & J' & S \\ J & K & 1 \end{Bmatrix} \begin{Bmatrix} J' & F' & I \\ F & J & 1 \end{Bmatrix} \begin{pmatrix} F' & 1 & F \\ -M & q & M \end{pmatrix}$$

by application of Edmond's 5.4.1 and 7.1.7 twice. Sometimes the first 6J symbol and the reduced matrix element are calculated together. In our case, we must be able to calculate the 6J and 3J symbols from the quantum numbers. There are several ways of doing this for the various types of 3J and 6J symbols: we present below (in group 13) a program in FOCAL for 3J symbols. This is normally used as part of a much larger program for calculating Hanle signal strengths in molecules, and so the ASK statement at the beginning is superfluous. Also, before a D 13 is executed, the program checks that the condition M' - q - M = 0 is met. That is, q is set equal to M'-M.

```
13.Ø1   A     ?EF?,?GF?,?EM?,?GM?
13.Ø2   S     SI=1
13.Ø3   IF    (EF-GF) 13.Ø4, 13.Ø5, 13.Ø5
13.Ø4   DO    13.7Ø; S SI=(-1)↑(GF+EF+1); S SJ=-1
13.Ø5   IF    (GM-EM+1) 13.Ø7, 13.Ø6, 13.Ø7
13.Ø6   DO    13.71; S SJ=SI*(-1)↑(GF+EF+1); S SS=-1
13.Ø7   IF    ( (EF-GF)↑2-1) 13.Ø8, 13.Ø8, 13.6Ø
13.Ø8   IF    ( (EM-GM)↑2-1) 13.13, 13.13, 13.61
13.13   IF    (EF-GF) 13.2Ø, 13.15, 13.2Ø
13.15   IF    (EM-GM) 13.3Ø, 13.33, 13.3Ø
13.2Ø   IF    (EM-GM) 13.36, 13.39, 13.36
13.3Ø   S     TJ=(EF-EM)*(EF+EM+1)*2/(EF+1)*(2*EF+1)*(4*EF)
13.31   S     TJ=(-1)↑(EF-EM)*FSQT(TJ); GOTO 13.45
13.33   S     TJ=EM↑2/(2*EF+1)*(EF+1)*(EF)
13.34   S     TJ=(-1)↑(EF-EM)*FSQT(TJ); GOTO 13.45
13.36   S     TJ=(EF-1-EM)*(EF-EM)/(2*EF+1)*(2*EF)*(2*EF-1)
13.37   S     TJ=(-1)↑(EF-EM)*FSQT(TJ); GOTO 13.45
13.39   S     TJ=(EF+EM)*(EF-EM)*2/(2*EF+1)*(2*EF)*(2*EF-1)
13.4Ø   S     TJ=(-1)↑(EF-EM)*FSQT(TJ); GOTO 13.45
13.45   S     TJ=TJ*SI; S SI=1; IF (SJ+1) 13.5Ø, 13.47, 13.5Ø
13.47   DO    13.7Ø
13.5Ø   S     SJ=1; IF (SS+1) 13.55, 13.52, 13.55
13.52   DO    13.71
13.55   S     SS=1;T ?TJ?,!;Q
13.60   T     "DELTA J IS BIGGER THAN 1, TRY AGAIN", !
13.61   T     "DELTA M IS BIGGER THAN 1, TRY AGAIN", !
13.7Ø   S     XX=GF; S GF=EF; S EF=XX; S XX=GM; S GM=EM; S EM=XX
13.71   S     GM=-GM; S EM=-EM
*
```

We also have written FOCAL programs for the 6J symbols as well as for part of the calculation of the reduced matrix elements, all to be used in the Hanle signal strength program. The entire program has more than 140 lines of FOCAL, and most of the lines have several instructions. It enables us to calculate many different quantum mechanical averages over all states allowed in particular set of transitions. We can average over light polarization or we can calculate separate sums for each polarization. We can easily change the nature of the weighting function to include additional effects (such as the Boltzmann population distribution in the ground state.)